Maximum and Minimum Values

Math 20

March 1, 2006

Purpose of this Document

This is a *Mathematica* notebook to reproduce the calculations we did in class today. Its goal is also to show how *Mathematica* can help you in future calculations

Example 1

We wanted to find and classify all critical points of the function

```
In[80]:= f = x^2 + x y + y^2 + 3 x - 3 y + 4
Out[80]= 4 + 3 x + x^2 - 3 y + x y + y^2
```

Note that "*x y*" needs to be entered with a space in between—otherwise it's the variable named xy.

■ Finding the critical points

```
In[81]:= D[f, x]
Out[81]= 3 + 2 x + y

In[82]:= D[f, y]
Out[82]= -3 + x + 2 y

In[83]:= Solve[{D[f, x] == 0, D[f, y] == 0}, {x, y}]
Out[83]= {{x → -3, y → 3}}
```

So we only have the single critical point

Second derivatives can be taken by putting the variable in the arguments as many times as you want to differentiate:

```
In[86]:= D[f, x]
Out[86]= 3 + 2 x + y
```

```
In[85]:= D[f, x, x]

Out[85]= 2
```

This is a little fanciness to take the derivatives of f with respect to each variable, twice.

```
In[87]:= H = Outer[D[f, #1, #2] &, {x, y}, {x, y}]

Out[87]= {{2, 1}, {1, 2}}

In[88]:= MatrixForm[H]

Out[88]//MatrixForm=
        ( 2  1 )
        ( 1  2 )

In[89]:= Det[H]

Out[89]= 3
```

So we can conclude the single critical point is a local minimum.

■ Plots to verify

```
In[92]:= Plot3D[f, {x, -4, 0}, {y, 0, 4}]
```



```
Out[92]= - SurfaceGraphics -
```

```
In[93]:= ContourPlot[f, {x, -4, 0}, {y, 0, 4}]
```




```
Out[93]= - ContourGraphics -
```

You can see in the contour plot the minimum at $x = -3$, $y = 3$.

Example 2

Here we are studying a more complicated function

```
In[94]:= f = 4 x y - x^4 - y^4
```

```
Out[94]= -x^4 + 4 x y - y^4
```

```
In[101]:=
        cp = Solve[{D[f, x], D[f, y]} == {0, 0}, {x, y}]
```

```
Out[101]=
        {{x → -1, y → -1}, {x → 0, y → 0}, {x → -i, y → i}, {x → i, y → -i},
         {x → 1, y → 1}, {x → -(-1)^(1/4), y → -(-1)^(3/4)}, {x → (-1)^(1/4), y → (-1)^(3/4)},
         {x → -(-1)^(3/4), y → -(-1)^(1/4)}, {x → (-1)^(3/4), y → (-1)^(1/4)}}
```

Mathematica gives us all nine of the roots here; we only want the real ones. We could just cut and paste them. Here's another way:

```
In[109]:=
        realcp = Select[cp, (x ∈ Reals && y ∈ Reals /. #) &]
```

```
Out[109]=
        {{x → -1, y → -1}, {x → 0, y → 0}, {x → 1, y → 1}}
```

In[112]:=
```
H = Outer[D[f, #1, #2] &, {x, y}, {x, y}];
MatrixForm[H]
```

Out[113]//MatrixForm=

$$\begin{pmatrix} -12\,x^2 & 4 \\ 4 & -12\,y^2 \end{pmatrix}$$

In[114]:=
```
H /. realcp // Map[MatrixForm, #] &
```

Out[114]=

$$\left\{\begin{pmatrix} -12 & 4 \\ 4 & -12 \end{pmatrix}, \begin{pmatrix} 0 & 4 \\ 4 & 0 \end{pmatrix}, \begin{pmatrix} -12 & 4 \\ 4 & -12 \end{pmatrix}\right\}$$

We can see that if $x = 1$ or $x = -1$, the critical point is a local maximum. At (0, 0), the critical point is a saddle.

In[119]:=
```
Plot3D[f, {x, -1.2, 1.2}, {y, -1.2, 1.2}]
```



Out[119]=
```
- SurfaceGraphics -
```

The surface plot is a little hard to see the critical points on. That's because large x and y cause f to get large negative quickly. The contour plot reveals a little more.

In[120]:=

```
ContourPlot[f, {x, -1.2, 1.2}, {y, -1.2, 1.2}]
```



Out[120]=

```
- ContourGraphics -
```